

## 目次

9 平面のグラフ	2
9.1 $y = f(x)$ のグラフ	2
9.1.1 $y = f(x)$ のプロットの基本	3
9.1.2 自分で関数を作る	6
9.1.3 継ぎはぎ関数のグラフ	10
9.1.4 Grid option	10
9.1.5 様々な $y = f(x)$ のグラフ (数 )	12
9.2 $f(x, y) = 0$ のプロット	19
9.2.1 基本のプロット	19
9.2.2 いくつかのグラフを一緒に描くとき	21
9.2.3 $f(x, y) = 0$ のグラフの応用	21
9.3 媒介変数 (パラメーター) 表示のグラフ	23
9.3.1 基本のプロット	23
9.3.2 2つ以上のグラフを一緒に描くとき	25
9.4 極座標のグラフ (数 C の範囲)	26
9.5 【参考】グラフの色の変更	28
9.6 【参考】異なった座標系で描かれたグラフの重ね合わせ	29
9.7 【参考】オプションのリスト (一部のみ)	30

## 9 平面のグラフ

$y = f(x)$ のグラフ ( $a \leq x \leq b$ )	<code>plotfunc2d(f(x), x = a ..b)</code>
$y = f(x)$ のグラフ ( $a \leq x \leq b$ )	<code>plot :: Function2d(f(x), x = a ..b)</code>
$f(x, y) = 0$ のグラフ ( $a \leq x \leq b, c \leq y \leq d$ )	<code>plot :: implicit(f(x, y), x = a ..b, y = c ..d)</code>
媒介変数表示のグラフ ( $a \leq t \leq b$ )	<code>plot :: Curve2d([x(t), y(t)], t = a ..b)</code>
極座標 $(r, \theta) = (f(t), g(t))$ のグラフ ( $a \leq t \leq b$ )	<code>plot :: polar([f(t), g(t)] t = a ..b)</code>

注1) MuPAD ではいろいろな形で表されたグラフを描くことができます。特に  $y = f(x)$  のグラフを描く `plotfunc2d` は簡単で、たくさんのグラフを一緒に描くことができます。そのうえ、MuPAD は、違った描写法で描かれたグラフを一緒に描写することも出来ます。さらに、いろいろなオプションがあり、グラフの描写の様子を細かく指定することができます。詳しくは、各節を見てください。

### 9.1 $y = f(x)$ のグラフ

$y = f(x)$  のグラフを、 $a \leq x \leq b$  の範囲で描くのは `plotfunc2d()`; を次のようにしてつかいます。<sup>注2)</sup>

```
plotfunc2d(f(x), x = a ..b);
```

$y$  の範囲も  $c \leq y \leq d$  のように指定したいときは次のようにします。

```
plotfunc2d(f(x), x = a ..b, y = c ..d);
```

2 つ以上のグラフを重ねて描きたいときは、コンマで区切るだけです。

```
plotfunc2d(f(x), g(x), x = a ..b);
```

scene option をつけるのは、次のようにコンマで区切って指定します。2 つ以上の scene option をつけるときは、コンマで区切って指定します。

```
plotfunc2d(option, f(x), x = a ..b);
plotfunc2d(option1, option2, ..., f(x), x = a ..b);
```

scene option というのは全体の描写を決めるオプションで、主なものとしては `Scaling`, `Ticks`, `BackGround`, `ForeGround`, `Axes`, `Arrows` などあり、それぞれの値を `Scaling=Constrained` (実際のスケールで表示), `Ticks=[20, 10]` ( $x$  軸方向の目盛りを 20 個,  $y$  軸方向の目盛りを 10 個刻む), `Title="2 ji kansuu"` (2 ji kansuu というタイトルをつける) のように (名前)=(値) のように指定します。とてもたくさんの種類の指定が出来ますが、私が良く使うのは `Scaling=Constrained` ぐらいで、たまに、`Ticks`, `Title` も使うくらいです。オプションに関しては、最後の節に簡単な表にしていますので、詳しくはそちらを見てください。<sup>注3)</sup>

注1) `plot::~` の形のコマンドは `plot(plot::~)` のように `plot()` と組み合わせて使う。

注2) `plotfunc2d` は、`plot+function+2d`(2次元の関数のグラフのプロット)の意味。また、`plot::Function2d()` のほうはグラフの色を変えたり、違う座標系で描かれたグラフを重ねるときに使います。後の節(違う座標系で描かれたグラフを重ね合わせの章)で、参考として扱います。

注3) MuPAD の Help では、scene option は  $f(x)$  の前につけることになっていますが、実際は `plotfunc2d(f(x), x = a ..b, option)`; のように、後につけても大丈夫でした。その他のオプションとして `plot option` といって `Color`(グラフの色), `Grid`(パラメータの刻み数) など、個々のグラフの描写の仕方を指定するオプションもあり、これらはつける位置が異なります。しかし、`plotfunc2d` では `plot option` は、`Grid` しか指定できません。

### 9.1.1 $y = f(x)$ のプロットの基本


#### (1) 多項式のグラフ

整関数 (多項式) のグラフで基本を習得しましょう。

$y = x^3 - 3x + 1$  ( $-3 \leq x \leq 3$ ) のグラフ

• `plotfunc2d(x ^ 3 - 3 * x + 1, x = -3..3);`

と打ちます。すると、別の window が開き次のようなグラフが見えるはずです。




もし window が開かないときは、スタートボタンの右のほうに "Graphics-Vcam..." というボタンが見えるはずです。これをクリックしてください。さて、いつも教科書で見ているグラフより y 軸方向につぶれて見えますね。これは MuPAD が自動的に y 軸方向の目盛り (Scale) を変更して書いているためです。(y 軸の '15' という目盛りで解ります。)

オプションで `Scaling=Constrained` <sup>注4)</sup> と指定してみます。


• `plotfunc2d(Scaling = Constrained, x ^ 3 - 3 * x + 1, x = -3..3);`

<sup>注4)</sup> Constrained と指定すると x,y 軸方向の目盛りのとり方が同じになります。Unconstrained と指定すると、MuPAD が自動的に目盛りを変更します。初期値では Unconstrained になっています。ちなみに 'constrain' とは '抑制する、押さえつける' という意味なので、Scaling=Constrained とは、MuPAD の scaling 機能を抑制するということなのではないでしょうか？



これが本当の姿ですが、少し見づらいですね。今度は、 $y$  軸方向の変域も  $-5 \leq y \leq 5$  に指定してみます。

• `plotfunc2d(x ^ 3 - 3 * x + 1, x = -3..3, y = -5..5);`




$-3 \leq x \leq 3$  かつ  $-5 \leq y \leq 5$  の範囲しか表示されません。 $y$  軸方向に拡大したいときはこのように  $y$  の範囲も指定するといいいでしょう。最後に、微分を使って結果を確かめて見ましょう。 $f(x) = x^3 - 3x + 1$  とおくと、 $f'(x) = 3x^2 - 3 = 3(x-1)(x+1)$

$x$	...	-1	...	1	...
$f'(x)$	+	0	-	0	+
$f(x)$	↗	3	↘	-1	↗

ちょっとグラフの目盛り刻みが大きすぎてよく解りませんね。このような時は Ticks<sup>注5)</sup> を使って,  $\text{Ticks} = [n_x, n_y]$ 、または  $\text{Ticks} = n$  ( $n_x = n_y = n$  のとき) と指定します。  $n$  が大きいほど刻みの数が増えます。

• `plotfunc2d(Ticks = 10, x ^ 3 - 3 * x + 1, x = -3..3, y = -5..5);`




これで極値をグラフから読み取れます。どうやら合っているみたいですが。他の関数でも、いろいろ”自分で”やってみましょう。(昔から自分(微分)の事は自分でやれといひます。これは、有名なジョーク? です。)

## (2) 指数・対数関数のグラフ

指数・対数関数のグラフも描いてみましょう。  $a^x, e^x, \log_a x, \log x$  は MuPAD では、それぞれ  $a ^ x, E ^ x$ (または  $\exp(x)$ ),  $\log(a, x), \ln(x)$  でした。  $y = 2^x, y = \log_2 x, y = x$  のグラフを一度に描いてみます。グラフを重ねて描くのは、単に関数をコンマで区切って書くだけです。また、  $\text{Scaling}=\text{Constrained}$  にして  $x, y$  軸方向の目盛りを同じにします。

• `plotfunc2d(Scaling = Constrained, 2^x, log(2, x), x, x = -5..5, y = -5..5);`

<sup>注5)</sup> tick というのは時計の tick, tick(チクタク)と同じで’刻み’という意味です。



$y = 2^x$  と  $y = \log_2 x$  のグラフが直線  $y = x$  に関し対称となっていることが解ります。

### 9.1.2 自分で関数を作る

次は三角関数のグラフ (単位は度) を描きましょう。しかし、MuPAD では角度の単位はラジアンになってしまうので、自分で新しく関数を定義しなくてははいけません。<sup>注6)</sup>  $y = f(x)$  と新しい関数を定義するには  $y := x \rightarrow f(x)$ ; とします。→ は、矢印「 $\rightarrow$ 」を表していて、 $x$  に  $f(x)$  を対応させるという感じです。さて、 $180^\circ = \pi$  ラジアンだから  $f(x) = \text{Sin}(x)$  (単位は度) を MuPAD の  $\text{sin}(x)$  (ラジアン単位) を使って定義すると  $\text{Sin}(x) = \text{sin}\left(\frac{x}{180} \times \pi\right)$ . よって


```
• Sin := x -> sin(x/180 * PI);          >> x -> sin(x/180 * PI)
```

これを使って  $y = \text{Sin}(x) (0^\circ \leq x \leq 360^\circ)$  を描くには

```
• plotfunc2d(Sin(x), x = 0..360);
```


と打つだけです。

<sup>注6)</sup> ラジアンは数 で習う。三角関数の章参照。また、関数の定義については、文字式の取り扱いの章を参照



今度は Scale をいじる必要はなさそうです。しかしグラフの上部をみると、タイトルが  $\sin(x/180 * \text{PI})$  となっています。これがいやならば、Title Option で Title="タイトル名" のように指定し、タイトルを変えます。このとき " " のように囲むのを忘れないようにしましょう。ここでは "sin(x)&cos(x)" というタイトルにします。<sup>注7)</sup> ついでに  $y = \cos(x)$  ( $0^\circ \leq x \leq 360^\circ$ ) のグラフと重ねて書いて見ます。コンマで区切って書くだけでしたね。


- `Cos := x -> cos(x/180 * PI);`
- `plotfunc2d(Title = "sin(x)&cos(x)", Sin(x), Cos(x), x = 0..360);`



次は  $\tan(x)$  ( $0^\circ \leq x \leq 360^\circ$ ) のグラフです。"tan(x)" というタイトルをつけました。


- `Tan := x -> tan(x/180 * PI);`
- `plotfunc2d(Title = "tan(x)", Tan(x), x = 0..360);`

<sup>注7)</sup> " " で囲むのは中が文章 (string) 型のデータであるという意味でコンピュータの世界では良く使われます。" "の中は何を書いてもエラーにはなりません。" kono naka ha jiyuu desu... "



$x \rightarrow 90^\circ$  のとき  $\tan(x) \rightarrow \pm\infty$  となるので、グラフは  $y$  軸方向に画面を突き抜けてしまいます。 $y$  軸方向の拡大は  $y$  の変域を指定すればよかったですね。 $-10 \leq y \leq 10$  に指定します。

• `plotfunc2d(Title = "tan(x)", Tan(x), x = 0..360, y = -10..10);`



今度はうまくいきました。


三角関数の合成はうまく出来ませんでしたが、(三角関数の章参照) グラフは描けるはずですよ。

$$\sin x + \sqrt{3} \cos(x) = 2 \sin(x + 60^\circ)$$

ですから、 $y = 2 \sin x$  と  $y = \sin x + \sqrt{3} \cos x$  のグラフを同時に描いてみましょう。

• `plotfunc2d(Sin(x) + sqrt(3) * Cos(x), 2 * Sin(x), x = 0..360);`





赤色のグラフと  $x$  軸との交点の1つが  $(300, 0)$  になっているので、 $y = \sin x + \sqrt{3} \cos x$  のグラフは  $y = 2 \sin x$  のグラフを  $x$  軸方向に  $-60^\circ$  平行移動したグラフとなっていることがわかります。

### 9.1.3 継ぎはぎ関数のグラフ


$$f(x) = \begin{cases} x < 0 \text{ のとき,} & -x \\ x \geq 0 \text{ のとき,} & x^2 \end{cases}$$

のような、いわゆる継ぎはぎ関数の定義には `piecewise([f1],[f2],...)` を使います。<sup>注8)</sup> 例えば、上の関数を `f` という名前をつけて定義するには、次のようにします。

• `f := piecewise([x < 0, -x],[0 <= x, x ^ 2]);`      `>> -x if x < 0, x^2 if 0 <= x`

ここで矢印 (`->`) を入れて、`x->piecewise` としないでください。また、MuPAD では `<`, `>`, `,` はそれぞれ `<`, `>`, `<=`, `>=` となるのでした。さて、この関数のグラフを、`-2 ≤ x ≤ 2` の範囲で描いてみましょう。

• `plotfunc2d(f(x), x = -2..2);`



### 9.1.4 Grid option

ここからは、数 III のグラフです。この後の節では、 $\sin x$ ,  $\cos x$  の角度の単位はラジアン (*rad*) とします。数 III のグラフでは、普通に描くとグラフが滑らかにならずに、'ぎざぎざ' になることがあります。このような時は `Grid option` を使って `Grid=n` ( $n$  は自然数) と指定します。 $n$  が大きいほどグラフが滑らかになります。`Grid option` を使うときは、コンマで区切って次のように入力します。<sup>注9)</sup>


`plotfunc2d( f(x), x = a ..b, Grid = n);`

<sup>注8)</sup> `piecewise=piece+wise` で `piece` は「一片、断片」という意味、`wise` は「~の点では」という意味で `moneywise`, `timewise` などのように使います。「利口な」という意味ではありません。念のため。

<sup>注9)</sup> `grid` というのは格子という意味です。MuPAD がグラフを描くとき、全ての  $x$  成分に関する  $f(x)$  の値を計算することは出来ません。いくつかの代表的な数字に対する値を計算し、それによって得られる点をつなげています。 $n$  が増えると、そのような点


では、 $y = \sin \frac{1}{x} (-2 \leq x \leq 2, x \neq 0)$  のグラフを描いてみましょう。

- `plotfunc2d(sin(1/x), x = -2..2);`



このようにぎざぎざになっています。次は `Grid=500` と指定してやって見ましょう。Title もつけておきました。

- `plotfunc2d(Title = "sin(1/x), Grid = 500", sin(1/x), x = -2..2, Grid = 500);`



だいぶ滑らかになりました。

---

の個数が増えます。最初の設定 (default) では、`Grid=100` です。MuPAD の Help では、`grid option` は  $f(x)$  の後につけることになっていますが、実際は `plotfunc2d(Grid = n, f(x), x = a ..b);` のように、前につけても大丈夫でした。

$\sin \frac{1}{x} = \pm 1 \iff \frac{1}{x} = (2n \pm \frac{1}{2})\pi \iff x = \frac{1}{(2n \pm \frac{1}{2})\pi}$  ( $n$  は整数) ですから  $x = 0$  の近くでは無数に  $-1$  と  $1$  の値をとり振動します。また,  $x = 0$  のときは, 関数は定義されませんが MuPAD は気にしないようです。

### 9.1.5 様々な $y = f(x)$ のグラフ (数 )


この節では, どんどんグラフを描いていきます。

#### 例題 1


$y = e^{-x} \sin x$  ( $0 \leq x \leq 4\pi$ ) のグラフの概略を描け。

$y = e^{-x} \sin x$  ( $0 \leq x \leq 4\pi$ ) のグラフを描いてみましょう。横軸の単位はラジアンですが,  $2\pi, 3\pi$  などと表示されませんから  $\pi \approx 3.14$  を使って, 自分で換算する必要があります。

• `plotfunc2d(exp(-x)*sin(x), x = 0..4 * PI);`



残念ながら,  $\pi \approx 3.14$  を横軸の単位にとることはできません。でも  $4\pi = 4 \times 3.14 = 12.56$  ですから少なくとも定義域はあっています。しかし  $x > 6$  ではほとんど潰れています。 $e^{-x}$  は非常に速く小さくなるので, このようになります。y 軸方向に拡大しようとして y の変域を指定してみても, 余りにつぶれ方が速いので, うまくいきません。ところが, このグラフは教科書などで見ると



のようなグラフになっていますが、これは実は  $y = e^{-0.3x} \sin x$  のグラフです。指数が  $0.3x$  となっているので、 $x = 4\pi \approx 12.5$  の近くでもつぶれていません。教科書などでは解りやすいようにデフォルメしてある訳です。

なお、上下に描いたグラフは  $y = \pm e^{-0.3x}$  のグラフです。  $-1 \leq \sin x \leq 1$  だから

$$-e^{-kx} \leq e^{-kx} \sin(x) \leq e^{-kx}$$

ここで左側等号は  $\sin(x) = -1 \iff x = \frac{3}{2}\pi, \frac{7}{2}\pi$ , のとき成立するので  $x = \frac{3}{2}\pi, \frac{7}{2}\pi$  のとき  $y = -e^{-kx}$  のグラフと  $y = e^{-kx} \sin x$  のグラフは接します。また、 $x = \frac{1}{2}\pi, \frac{5}{2}\pi$  のとき  $y = e^{-kx}$  のグラフと  $y = e^{-kx} \sin x$  のグラフは接します。(上図)

### 例題 2


$y = x$  のグラフと  $y = \sqrt{4 - x^2}$  のグラフを利用して、 $y = x + \sqrt{4 - x^2}$  ( $-2 \leq x \leq 2$ ) のグラフを描け。

$y = \sqrt{4 - x^2}$  は半円を表し、 $y = x$  は直線を表すので、2つのグラフの  $y$  成分を加えていけば、微分しなくともグラフの概形は解ります。MuPAD を使って、3つのグラフを一度に描いてみましょう。なお、Scaling=Constrained を使って実寸法で表示し、Ticks=[5,10] を使って、 $x$  軸方向の目盛りを 5、 $y$  軸方向の目盛りを 10 に変えています。

- plotfunc2d(Scaling = Constrained, Ticks = [5, 10],  
x, sqrt(4 - x ^ 2), x + sqrt(4 - x ^ 2), x = -2..2);

グラフは次のようになります。

$$x(4-x^2)^{1/2} + (4-x^2)^{1/2}$$



増減を調べなくてもグラフが描けるような例のみを扱ってきましたが、今度は微分が必要なグラフです。


例題 3

$y = f(x) = \frac{x^3}{x^2 - 1}$  の増減をしらべてグラフを描け。また  $y = x$  のグラフとの関係を調べよ。

とりあえず、MuPAD で  $y = f(x)$  と  $y = x$  のグラフをまとめて描きましょう。

• `plotfunc2d(x, x ^ 3/(x ^ 2 - 1), x = -5..5, y = -5..5);`

次のようになります。



手計算で、結果を確認しておきましょう。

定義域は  $x \neq \pm 1$ . このとき、

$$y' = \frac{3x^2(x^2 - 1) - x^3 \cdot 2x}{(x^2 - 1)^2} = \frac{x^2(x + \sqrt{3})(x - \sqrt{3})}{(x^2 - 1)^2}$$

$y'$  の分母は常に正なので、 $f'(x)$  の符号は分子の符号と一致する。よって増減表は次のようになる。

$x$	$-\infty$	$\dots$	$-\sqrt{3}$	$\dots$	$-1$	$\dots$	$0$	$\dots$	$1$	$\dots$	$\sqrt{3}$	$\dots$	$\infty$
$y'$		$+$	$0$	$-$	$-$	$0$	$-$	$-$	$-$	$0$	$+$		
$y$	$-\infty$	$\nearrow$	$-\frac{3\sqrt{3}}{2}$	$\searrow$	$-\infty$   $\infty$	$\searrow$	$0$	$\searrow$	$-\infty$   $\infty$	$\searrow$	$\frac{3\sqrt{3}}{2}$	$\nearrow$	$\infty$

また割り算を実行して

$$\frac{x^3}{x^2 - 1} = x + \frac{x}{x^2 - 1}$$

$\lim_{x \rightarrow \pm\infty} \frac{x}{x^2 - 1} = 0$  だから、 $\lim_{x \rightarrow \pm\infty} \{f(x) - x\} = 0$  よって、 $x \rightarrow \pm\infty$  のとき、 $y = f(x)$  は  $y = x$  に限りなく近づく。  
( $y = x$  は漸近線)


確かに正しいようです。

例題 4


$y = f(x) = \sqrt[3]{x^3} (-2 \leq x \leq 2)$  のグラフを MuPAD で描け。

次のようにするとどうでしょう？

- `plotfunc2d((x ^ 3) ^ (1/3), x = -2..2);`



これだと上図のように  $x \geq 0$  の範囲でしか図示されません。しかし、実際は  $x$  が実数のとき  $\sqrt[3]{x^3} = x$  が成



なぜでしょう？ MuPAD では、 $a > 0$  のとき、 $\sqrt[3]{a} = a^{1/3}$  と表しました。しかし、 $a < 0$  のときは  $\sqrt[3]{a}$  は  $a$  の虚数 3 乗根を現してしまいます。<sup>注10)</sup> この例では、 $x < 0$  のとき  $x^3 < 0$  となるので、 $x^{1/3}$  は  $x < 0$  でプロットできません。

<sup>注10)</sup> 例えば `• rectform((-1) ^ (1/3); >> 1/2 + 3/2 I` になります。ちなみに、`rectform` は  $p + qi$  ( $p, q$  実数) の形にするコマンドでした。




例題 5

$y = f(x) = \sqrt[3]{x^3 - x^2}$  ( $-2 \leq x \leq 2$ ) の増減をしらべてグラフを描け。また漸近線も求めよ。

とりあえず, MuPAD で  $y = f(x)$  のグラフを描きましょう。MuPAD で  $a > 0$  のとき,  $a$  の 3 乗根は  $a^{\frac{1}{3}}$  と表しました。しかし, 今度は  $x^3 - 3x$  が常に正ではないので注意が必要です。

• `plotfunc2d((x ^ 3 - x ^ 2) ^ (1/3), x = -2..2);`

しかし, これでは下図のようになってしまいます。



これは MuPAD が  $x^3 - x^2 < 0$  の区間をプロットしていないからです。MuPAD では  $a < 0$  のとき,  $\sqrt[3]{a}$  で虚数 3 乗根をもとめてしまうので  $x^3 - x^2 < 0$  となる区間ではプロットできません。

そこで,

$$\sqrt[3]{a} = \begin{cases} a^{\frac{1}{3}} & (a > 0 \text{ のとき}) \\ 0 & (a = 0 \text{ のとき}) \\ -(-a)^{\frac{1}{3}} & (a < 0 \text{ のとき}) \end{cases} \quad \text{例 ; } \begin{cases} \sqrt[3]{8} = 8^{\frac{1}{3}} = 2 \\ \sqrt[3]{-8} = -(8^{\frac{1}{3}}) = -2 \end{cases}$$

を使います。  $x^3 - x^2 = x^2(x - 1) > 0 \iff x > 1$ ,  $x^3 - x^2 = x^2(x - 1) = 0 \iff x = 1$  ですから次のように継ぎはぎ関数で定義すると良いでしょう。

$$\sqrt[3]{x^3 - x^2} = \begin{cases} (x^3 - x^2)^{\frac{1}{3}} & (x > 1 \text{ のとき}) \\ -(x^2 - x^3)^{\frac{1}{3}} & (x < 1 \text{ のとき}) \\ 0 & (x = 1 \text{ のとき}) \end{cases}$$


したがって MuPAD では次のようになります。(  $x = 0$  のときの定義は, 省略できます)

• `f := piecewise([x >= 1, (x ^ 3 - x ^ 2) ^ (1/3)], [x < 1, -(x ^ 2 - x ^ 3) ^ (1/3)]);`  
`>> piecewise((-x ^ 2 + x ^ 3)^{\frac{1}{3}} if 1 <= x, -(x ^ 2 - x ^ 3)^{\frac{1}{3}} if x < 1)`

続けてプロットします。

• `plotfunc2d(Title = "(x ^ 3 - x ^ 2)no 3joukon", x = -1/3, f(x), x = -2..2);`

今度は、次のようなグラフになります。(ここでは  $y = x - \frac{1}{3}$  のグラフも一緒に描いています。理由はあとでわかります。) またタイトルもつけました。



では、手計算で確認しましょう。 $x < 0$  のときでも、 $x^{\frac{1}{3}} = \sqrt[3]{x}$  と定義すると、 $0$  以外の実数に対して  $(x^{\frac{1}{3}})' = \frac{1}{3}x^{-\frac{2}{3}}$  が成り立ちますから、

$$y = (x^3 - x^2)^{\frac{1}{3}}$$

よって、 $x^3 - x^2 \neq 0 \iff x \neq 0, x \neq 1$  のとき

$$y' = \frac{1}{3}(x^3 - x^2)^{-\frac{2}{3}} \cdot (3x^2 - 2x) = \frac{x(3x - 2)}{3\sqrt[3]{(x^3 - x^2)^2}} = \frac{3x - 2}{3\sqrt[3]{x(x-1)^2}}$$

$x$	...	0	...	$\frac{2}{3}$	...	1	...
$y'$	+		-	0	+		+
$y$	↗	0	↘	$-\frac{\sqrt[3]{4}}{3}$	↗	0	↗

また  $\sqrt[3]{x^3 - x^2} = \sqrt[3]{(x - \frac{1}{3})^3 - \frac{1}{3}x + \frac{1}{27}} = (x - \frac{1}{3}) \sqrt[3]{1 + \frac{-\frac{1}{3}x + \frac{1}{27}}{(x - \frac{1}{3})^3}}$  だから  $x \rightarrow \pm\infty$  のとき、 $y = \sqrt[3]{x^3 - x^2}$  は限りなく  $y = x - \frac{1}{3}$  に近づく。よって、漸近線は  $y = x - \frac{1}{3}$ 。これは確かにグラフより確かめられる。

## 9.2 $f(x,y) = 0$ のプロット

円の方程式 ;  $x^2 + y^2 - r^2 = 0$  のように  $f(x,y) = 0$  の形をした関数は ,  $y = f(x)$  の形に直して plot-func2d をつかうか , または plot::implicit を使って描きます。<sup>注11)</sup> 例えば ,  $f(x,y) = 0$  のグラフを , 領域 ( $a \leq x \leq b, c \leq y \leq d$ ) の中で描かせるには plot() と一緒にして次のように入力します。

- plot(plot :: implicit(f(x,y), x = a..b, y = c..d));

いくつかのグラフを重ねたいときは , コンマで区切って [] の中に入れます。

- plot(plot :: implicit([f(x,y), g(x,y)], x = a..b, y = c..d));

scene option(Scaling=Constrained など) をつけたいときは , plot の () の中に入れます。

- plot(  
plot :: implicit(f(x,y), x = a..b, y = c..d)  
, scene option);

<sup>注12)</sup> plot option(Grid=500 ,Color=RGB::Blue など) をつけたいときは , plot::implicit() の括弧の中に入れます。これは plot option は個々のグラフの描写に関する事なので当然ですね。

- plot (plot :: implicit(f(x,y), x = a..b, y = c..d, plot option));

### 9.2.1 基本のプロット

まずは円を描きましょう。  $x^2 + y^2 = 1 \iff x^2 + y^2 - 1 = 0$  ですから次のようにします。変域は  $-1 \leq x \leq 1, -1 \leq y \leq 1$  を含んでいれば大丈夫です。また、入力が長くなって見づらい時は「Shift キー」を押しながら「Enter キー」を押すとエラーを出さずに改行できます。


- plot(  
plot :: implicit(x ^ 2 + y ^ 2 - 1, x = -1..1, y = -1..1)  
•);

このとき、 $x ^ 2 + y ^ 2 = 1$  とうっかりやっけてしまいがちなので注意してください。さて結果は

---


<sup>注11)</sup>  $f(x,y) = 0$  で表される関数を陰関数 (implicit function) といいます。これに対し  $y = f(x)$  で表される関数を陽関数 (explicit function) といいます。plot::implicit というのは plot library に入っている implicit コマンドのことでこのライブラリには、プロットのコマンドがたくさん入っています。

<sup>注12)</sup> 解りやすくするため 3 行に分けましたが、もちろん 1 行に入力してもいいです。何行かに分けて入力するときは Shift key を押しながら、Return key を押してください。



円がつぶれてしまいました。これは MuPAD が自動的に目盛り (scale) を調整しているからでした。scene option で Scaling=Constrained と Title="x ^ 2+y ^ 2=1" をつけてもう一度やってみましょう。<sup>注13)</sup>

- plot(
- plot :: implicit(x<sup>2</sup> + y<sup>2</sup> - 1, x = -1..1, y = -1..1)
- , Scaling = Constrained, Title = "x ^ 2 + y ^ 2 = 1");




<sup>注13)</sup> MuPAD のヘルプでは scene option は plot::implicit() の後につけることになっていますが、実際には前につけても大丈夫みたいです。

### 9.2.2 いくつかのグラフを一緒に描くとき

楕円と双曲線を一緒に書いて見ます。plotfunc2d() のときと異なり、この場合 [] で囲まないといけません。なお、Ticks=10 で目盛り刻みの数を 10 にしました。

- plot(
- plot::implicit([x ^ 2/4 + y ^ 2 - 1, x ^ 2/4 - y ^ 2 - 1, x ^ 2/4 - y ^ 2 + 1], x = -5..5, y = -5..5)
- , Ticks = 10);



それぞれ順に

$$\frac{x^2}{2^2} + y^2 = 1, \frac{x^2}{2^2} - y^2 = 1, \frac{x^2}{2^2} - y^2 = -1,$$

と変形できますから、たしかに上の様になります。

### 9.2.3 $f(x, y) = 0$ のグラフの応用

以前 (2001 年 7 月号の「大学への数学」学力コンテスト 6 番) に、採用された自作問題です。決して難しい問題ではないので、皆さんもやってみてください。<sup>注14)</sup>

#### 例題

実数  $x, y$  ( $0 < x < \frac{\pi}{2}, 0 < y < \frac{\pi}{2}$ ) が  $\cos x + 2 \cos y = 2$  … (\*)

をみたしている。このとき、次の問いに答えよ。

- (1) 増減、凹凸を調べて、(\*) のグラフを  $xy$  平面に描け。
- (2)  $x, y$  が (\*) の条件をみたすとき、 $x + y$  の最大値を  $\theta_0$  とする。  $\cos \theta_0$  の値を求めよ。

<sup>注14)</sup> 残念ながら、私は大学への数学の編集部とは直接の関係はありません。時々、問題を作って送っているだけです。

これを何人かの生徒にやらせると、以外にも(1)のほうで間違っていました。正解は次のようになります。


【解答】(1) (\*) の両辺を  $x$  で微分して、

$$-\sin x - 2 \sin y \cdot \frac{dy}{dx} = 0 \quad \frac{dy}{dx} = -\frac{\sin x}{2 \sin y} < 0 \quad \dots \textcircled{1}$$

①の両辺を  $x$  で微分して

$$\begin{aligned} \frac{d^2y}{dx^2} &= \frac{d}{dx}(-\sin x) \cdot \frac{1}{2 \sin y} + (-\sin x) \cdot \frac{d}{dy} \left( \frac{1}{2 \sin y} \right) \cdot \frac{dy}{dx} \\ &= \frac{-\cos x}{2 \sin y} + \sin x \cdot \frac{\cos y}{2 \sin^2 y} \cdot \left( -\frac{\sin x}{2 \sin y} \right) < 0 \end{aligned} \quad \dots \textcircled{2}$$

①,② より、グラフは減少関数で上に凸だから、左図のようになる。



(2)  $l: x+y=k$  において、 $l$  と (1) のグラフが共有点を持つような最大の  $k$  が  $\theta_0$  となる。 $l$  は傾き  $(-1)$  の直線だから、接点を  $P(x,y)$  とすると、

$$\frac{dy}{dx} = -1 \iff -\frac{\sin x}{2 \sin y} = -1 \iff \sin x = 2 \sin y \quad \dots \textcircled{3}$$

$P$  は曲線上にあるから

$$\cos x + 2 \cos y = 2 \quad \dots \textcircled{4}$$

③,④ より  $x,y$  は右図の角となるので、余弦定理より

$$\cos(\pi - (x+y)) = \frac{1^2 + 2^2 - 2^2}{2 \cdot 1 \cdot 2} = \frac{1}{4}$$


$\theta_0 = x+y$  だから

$$\cos \theta_0 = -\frac{1}{4} \quad \dots (\text{答})$$

$f(x,y) = 0$  のグラフを書くことによって、 $f(x,y) = 0$  という条件のもとで、いろいろな関数の最大・最小値が求まることがわかるといいます。さて、 $f(x,y) = 0$  のグラフの重要さが解ったところで、次は MuPAD でグラフを描いてみましょう。 $\sin x + 2 \cos y = 2$  のグラフを  $0 \leq x \leq \frac{\pi}{2}, 0 \leq y \leq \frac{\pi}{2}$  の範囲で書くのは、次のようにプロットします。

- plot(
- plot :: implicit(cos(x) + 2 \* cos(y) - 2, x = 0..PI/2, y = 0..PI/2)
- , Title = "cosx + 2cosy = 2");

タイトルもつけてみました。これでこうなります。



$\pi$  3.14 ですから、確かに 2 つのグラフは一致します。

### 9.3 媒介変数 (パラメーター) 表示のグラフ

$x = f(t), y = g(t) (a \leq t \leq b)$  と媒介変数表示されたグラフの描き方は `plot::Curve2d()` を `plot()` と一緒に使って次のようにします。

- `plot(plot :: Curve2d([f(t), g(t)], t = a..b));`

です。<sup>注15)</sup> オプションの指定の仕方は  $f(x, y) = 0$  のグラフのときと同様です。しかし、いくつかのグラフを一緒に描きたいときは、

- `plot(  
plot :: Curve2d([f1(t), g1(t)], t = a..b)  
plot :: Curve2d([f2(s), g2(s)], s = c..d)  
);`


のように描かないといけません。

#### 9.3.1 基本のプロット


手始めに円 ( $x^2 + y^2 = 4$ ) のパラメータ表示;  $x = 2 \cos t, y = 2 \sin t (0 \leq t \leq 2\pi)$  をやってみます。

- `plot(  
plot :: Curve2d([2 * cos(t), 2 * sin(t)], t = 0..2 * PI)  
• , Scaling = Constrained);`

<sup>注15)</sup> `Curve2d` は 2 次元の曲線の意味です。また、MuPAD では組み合わせが問題のときは、集合を表す記号 `{ }` を使い、順序が問題のときは、リストを表す記号 `[ ]` を用います。媒介変数表示の場合、 $x$  成分と  $y$  成分を逆には出来ませんからリストになります。



次はサイクロイド;  $x = t - \sin(t), y = 1 - \cos(t)$  ( $0 \leq t \leq 2\pi$ ) です。これは半径 1 の円を  $x$  軸の上に転がしたときに出来る図形です。




上図からサイクロイドのパラメータ表示は、次のようになります。

$$\begin{cases} x = t - \sin t \\ y = 1 - \cos t \end{cases}$$

これを Mupad で描いてみましょう。

- plot(
- plot :: Curve2d([t - sin(t), 1 - cos(t)], t = 0..2 \* PI)
- , Scaling = Constrained);





$2\pi = 6.28\dots$  なので大体あっているようです。

ついでに、 $x$  軸との間で囲まれた面積  $S$  を出してみましょう。 $x = t - \sin t$  とおくと  $\frac{dx}{dt} = 1 - \cos t$ ,  $\frac{x}{t} \Big|_0^{2\pi} \rightarrow \frac{2\pi}{2\pi}$  で  
すから

$$\begin{aligned}
 S &= \int_0^{2\pi} y \, dx = \int_0^{2\pi} y \frac{dx}{dt} dt = \int_0^{2\pi} (1 - \cos t)^2 dt \\
 &= \int_0^{2\pi} (1 - 2\cos t + \cos^2 t) dt \\
 &= \int_0^{2\pi} \left(1 - 2\cos t + \frac{1 + \cos 2t}{2}\right) dt \\
 &= \left[\frac{3}{2}t - 2\sin t + \frac{1}{4}\sin 2t\right]_0^{2\pi} \\
 &= 3\pi
 \end{aligned}$$

となるはずですが。実際、MuPAD でやると、

```
•int((1 - cos(t)) * diff(t - sin(t), t), t = 0..2 * PI);      >> 3PI
```


う～ん、MuPAD のほうが速い！

### 9.3.2 2つ以上のグラフを一緒に描くとき

plotfunc2d や plot::implicit と異なり、Curve2d それ自体では、2つ以上のグラフを一緒にには表示できません。しかし、plot() は幾つものグラフを重ねて表示できるので、それを利用します。<sup>注16)</sup> 例として、アステロイド； $x = \cos^3 t, y = \sin^3 t (0 \leq t < 2\pi)$  のグラフと円； $x = \cos t, y = \sin t (0 \leq t < 2\pi)$  のグラフを一緒に図示してみましょう。

- plot(
- plot :: Curve2d([cos(t) ^ 3, sin(t) ^ 3], t = 0..2 \* PI),
- plot :: Curve2d([cos(t), sin(t)], t = 0..2 \* PI),
- Title = "cycloid to enn", Scaling = Constrained)

<sup>注16)</sup> 実は plot() は、異なる座標系で描かれたグラフを重ねることも出来る。



#### 9.4 極座標のグラフ (数 C の範囲)

極座標が  $(r, \theta) = (f(t), g(t))$  のグラフは

- `plot(plot :: polar([f(t),g(t)], t = a ..b) );`

と入力します。特に  $r = f(\theta)$  のグラフ (教科書はこのタイプのみ) は


- `plot(plot :: polar([f(t), t], t = a ..b) );`

と入力します。オプションの指定は、 $f(x,y) = 0$  のグラフ、媒介変数表示のグラフと同様です。いくつかのグラフをまとめて描くときは、媒介変数表示のときと同様に `plot()` と組み合わせます。

極座標の復習です。  $r$  を点  $P(x,y)$  と原点との距離、 $\theta$  を点  $P$  の偏角とすると、極座標は  $(r, \theta)$  で定義され、直交座標  $(x,y)$  と

$$\begin{cases} x = r \cos t \\ y = r \sin t \end{cases}$$


の関係で結びつきます。(右図)



また、例えば右図の点 A,B,C の極座標はそれぞれ

$$A\left(\sqrt{2}, \frac{\pi}{4}\right), B\left(2, \frac{\pi}{6}\right), C\left(\sqrt{2}, -\frac{\pi}{4}\right)$$


となります。




極座標  $(r, \theta)$  間の関係式を極方程式といいます。例えば右図において中心が  $A(1,0)$ 、半径が 1 の円の極方程式は  $OP = OB \cos \theta$  より

$$r = 2 \cos \theta$$

となります。これを MuPAD で描くには次のようにします。



- plot(Scaling = Constrained,
- plot :: polar([2 \* cos(t), t], t = -PI/2..PI/2)
- );




## 9.5 【参考】グラフの色の変更

MuPAD では自動的にグラフの色が変化します。これを変えるのは、plot option の 1 つである Color をつかって、Color=[r,g,b]( r,g,b は 0 以上 1 以下の数) と RGB 指定するか、または Color=RGB::Black, Color=RGB::Yellow のようにカラー変数を用いて指定します。

注17)

しかし、Color option は、plot option の 1 つなので、残念ながら plotfunc2d では使えません。<sup>注18)</sup> ここで、 $y = f(x)$  のグラフの色を変えたいときは、より基本のコマンドである plot::Function2d() を使います。plot::Function2d の使い方は、plot::Curve2d や plot::polar などと同じです。例えば、 $y = x^2$  のグラフを黄色で描くには次のようにします。

- plot(  
• plot::Function2d(x ^ 2, x = -2..2, Color = RGB :: Yellow)  
• ,Scaling = Constrained);



このように plot option(Color,Grid など) は、個々のグラフの描写にかかわるオプションなので plot::~ の括弧の中に入れます。一方,scene option (Scaling, Ticks など) は全体の描写にかかわるオプションなので plot() のほうの括弧の中に入れます。う～ん、うまく出来てますね。

なお、plot::implicit の場合は Color でなく、Colors を使って Colors = [[Color<sub>1</sub>],[Color<sub>2</sub>],...] のように指

注17) RGB 指定とは光の 3 原色 (Red,Green,Blue) の混合比を指定したもので例えば、[1.0,0,0]=Red, [0,1,0]=Green,[0,0,1]=Blue ,[1,1,0]=Yellow, [1,1,1]=Black で、0 以上 1 以下の数で指定します。頭がデジタルでない人は、カラー変数を用いると楽でしょう。カラー変数は RGB::Black, RGB::White, RGB::Green, RGB::Red, RGB::Blue, RGB::Yellow, RGB::Gray, RGB::Magenta, RGB::Olive, RGB::LightGray, RGB::OliveGreenDark(どんな色じゃ?) など本当に”色々”ある。

注18) plotfunc2d で使えるのは scene option と grid のみ

定します。例えば、 $x^2 - y = 0$  のグラフを `plot::implicit` を使って黄色で描くには次のようにします。

- `plot(`
- `plot :: implicit(x ^ 2 - y, x = -2..2, y = -2..4, Colors = [RGB :: Yellow])`
- `, Scaling = Constrained);`


そういえば、`plot::implicit` を使って2つ以上のグラフを一緒に描くときは `[]` のなかに `[f1(x, y), f2(x, y)]` のように入れないといけなかったですね。Color も同じです。

## 9.6 【参考】異なった座標系で描かれたグラフの重ね合わせ

`plot()` コマンドを使うと、いままで異なった座標系を用いて描かれたグラフを重ね合わせることが出来ます。また自分でグラフの色を変えることも出来ます。ただしこのときは `plotfunc2d()` の代わりに、`plot::Function2d` を利用しないとイケません。(前節参照) 例として、媒介変数表示の円のグラフ ( $x = \cos t, y = \sin t$ ) と陽関数表示の放物線のグラフ ( $y = x^2$ ) を組み合わせてます。<sup>注19)</sup> さらに、このとき円のグラフを黄色、放物線を青色で描かせて見ます。

- `plot(`
- `plot :: Curve2d([cos(t), sin(t)], t = 0..2 * PI,`  
`Color = RGB :: Yellow),` … plot option
- `plot :: Function2d(x^2, x = -2..2, Color = RGB :: Blue)` … plot option
- `, Scaling = Constrained);` … scene option

これで次のようになります。



ここで、`Scaling=Constrained` は全体の描写に働くので、scene option と呼ばれ `plot()` の括弧の中に入れます。一方 `Color, Grid` などのオプションは個々のグラフの描写にかかわるオプションなので plot option と言いい、`plot::~` のほうの括弧の中に入れるのでした。

<sup>注19)</sup> もちろんこれは放物線のグラフを  $x = t, y = t^2$  と媒介変数表示すれば済むことなのですが、あくまで例としてやってみます。

## 9.7 【参考】オプションのリスト (一部のみ)

MuPAD のオプションは 全体のグラフに関するオプション (Scene option) と 個々のグラフ (object) に関するオプション (plot option) の 2 種類があり, それぞれつける位置も異なります。詳しくは Help BrouseMannal を見てください。

### 1.Scene options

Option 名	値	初期値	働き
Arrows	TRUE,FALSE	FALSE	x,y 軸の矢印
Axes	Box,Corner,None,Origin	Origin	x,y 軸の表示の仕方
BackGround	RGB::カラー名	RGB::White	背景色
ForeGround	RGB::カラー名	RGB::Black	前景色 (x,y 軸の色)
LineStyle	SolidLines,DashedLines	SolidLines	グラフの線種 (実線、点線)
Scaling	Constrained,Unconstrained	Unconstrained	両軸方向の目盛りのとり方
Ticks	Automatic, None , $[n_x, n_y]$ または $n$ ,	Automatic	x,y 軸方向の目盛り刻みの数
Title	”タイトル名”	入力した関数	タイトル
TitlePosition	Above,Below	Above	タイトルの位置

### 2 . plot options

Option 名	値	初期値	働き
Color	[R,G,B]		各グラフの色の指定
Grid	[n]	[100]	パラメータの分割数
Title	”タイトル名”	” ”	各グラフのタイトル
TitlePosition	[x,y]		各タイトルの位置

BackGround,ForeGround の色の指定には, カラー変数を用いると楽です。カラー変数は RGB::Black, RGB::White, RGB::Green, RGB::Red, RGB::Blue, RGB::Yellow, RGB::Gray, RGB::Magenta, RGB::Olive, RGB::LightGray, RGB::OliveGreenDark(どんな色じゃ?) など ”色々”ある。